

WOLFRAM KOEPF

Eine Vorstellung von MATHEMATICA und Bemerkungen zur Technik des Differenzierens

Arthur Engel hat in [1] die Fähigkeiten des PC-Programms DERIVE [4] für Mathematiklehrer vorgestellt.

Wir wollen in dieser Arbeit das Computeralgebrasystem MATHEMATICA [5] vorstellen. MATHEMATICA unterstützt regelorientiertes Programmieren, mit dem sich komplizierte Prozeduren so programmieren lassen, wie man sie in der Praxis auch durchführt.

Am Beispiel der Differentiation wollen wir diese Technik vorführen. In wenigen Schritten gelingt es uns, eine MATHEMATICA-Funktion `diff[f, x]` zu definieren, die die Differentiation des Ausdrucks f bezüglich der Variablen x vornimmt, d.h. die partielle Ableitung von f bzgl. x bestimmt.

Wenngleich MATHEMATICA bereits eine eingebaute Funktion `D[f, x]` zu diesem Zweck besitzt, ist es doch sehr lehrreich zu sehen, wie die üblichen Ableitungsregeln sich sukzessive zu einem Regelwerk verdichten, mit dem man in der Lage ist, jede durch algebraische Operationen aus elementaren Funktionen zusammengesetzte Funktion zu differenzieren.

1 MATHEMATICAS Fähigkeiten

MATHEMATICA ist ein interaktives Computeralgebrasystem, das numerische, graphische und symbolische Fähigkeiten hat, und in dem man prozedurale, listen- oder regelorientierte Programme schreiben kann.

MATHEMATICAS Fähigkeiten umfassen die folgenden Themenbereiche:

- rational exakte Berechnungen,
- beliebig genaue reelle Arithmetik,
- Arithmetik mit Polynomen (Expansion und Faktorisierung),
- Arithmetik mit rationalen Funktionen (Faktorisierung und Partialbruchzerlegung),
- symbolische Lösung von Polynomgleichungen,
- symbolische Lösung linearer Gleichungssysteme,
- Matrizenalgebra,
- Berechnung von Grenzwerten,
- Berechnung von Ableitungen,
- Berechnung von Taylorpolynomen,
- Berechnung von Stammfunktionen und bestimmten Integralen,
- symbolische Lösung gewöhnlicher Differentialgleichungen,
- sowie graphische Fähigkeiten.

In diesem Abschnitt geben wir einige Anwendungsbeispiele in einer typischen interaktiven MATHEMATICA-Sitzung. Hierbei ist `In[n]` die n -te Eingabezeile, welche vom Benutzer eingegeben und von MATHEMATICA durchnummeriert werden, während mit `Out[n]` die entsprechende Ausgabezeile MATHEMATICAS bezeichnet wird.

Beginnen wir mit der rational exakten Arithmetik. MATHEMATICA kennt die Fakultätsfunktion $n! = 1 \cdot 2 \cdot \dots \cdot n$ und die Eingabe von `50!` liefert

```
In[1]:= 50!
```

```
Out[1]= 30414093201713378043612608166064768844377641568960512000000000000
```

```
In[2]:= 1024/%
```

```
Out[2]= -----
                1
29701262892298220745715437662172625824587540594688000000000000
```

Hierbei steht das `/`-Zeichen für eine Division und das `%`-Zeichen für das Ergebnis der letzten Ausgabezeile. Mit `%n` wird die Ausgabe der n -ten Zeile abgekürzt. Man beachte, daß MATHEMATICA den Bruch automatisch gekürzt hat.

Mit `N[x]` erhält man den numerischen Wert von x . Die Kreiszahl π heißt bei MATHEMATICA `Pi`. Ihre Dezimaldarstellung ist also

```
In[3]:= N[Pi]
```

```
Out[3]= 3.14159
```

Eine 50-stellige Approximation dieser Zahl erhält man mit Hilfe des Aufrufs

```
In[4]:= N[Pi,50]
```

```
Out[4]= 3.1415926535897932384626433832795028841971693993751
```

MATHEMATICA kann auch symbolisch gegebene Objekte bearbeiten. So kann man z.B. Polynome eingeben und umformen. Die Funktion `Expand` expandiert einen Ausdruck, d.h. multipliziert alle Produkte aus.

```
In[5]:= f=Expand[Product[Product[j*x-y^k,{k,1,3}],{j,1,2}]]
```

```
Out[5]= 8 x6 - 12 x5 y + 4 x4 y2 - 12 x5 y2 + 18 x4 y3 - 12 x5 y3 - 6 x3 y4 +
> 22 x4 y4 - 12 x3 y5 + 18 x4 y5 + 2 x2 y6 - 27 x3 y6 + 4 x4 y6 +
> 9 x2 y7 - 12 x3 y7 + 11 x2 y8 - 6 x3 y8 - 3 x9 y2 + 9 x2 y9 - 3 x10 y +
> 2 x2 y10 - 3 x11 y12 + y
```

Hierbei weisen wir dem Symbol `f` als Wert diesen ausmultiplizierten Produktausdruck

$$\begin{aligned} \text{Product}[\text{Product}[j * x - y^k, \{k, 1, 3\}], \{j, 1, 2\}] &= \prod_{j=1}^2 \left(\prod_{k=1}^3 (jx - y^k) \right) \\ &= \prod_{j=1}^2 \left((jx - y)(jx - y^2)(jx - y^3) \right) \\ &= (x - y)(x - y^2)(x - y^3)(2x - y)(2x - y^2)(2x - y^3) \end{aligned}$$

zu. y^k bedeutet y^k und `Product[f, {k, 1, 3}]` berechnet das Produkt von f , wobei k die Zahlen 1 bis 3 durchläuft. Argumente von Funktionsaufrufen erfolgen in MATHEMATICA grundsätzlich in eckigen Klammern ([,]), um diese von (runden) Klammern in algebraischen Ausdrücken unterscheiden zu können. Mit geschweiften Klammern ({,}) werden Listen bezeichnet.

Eine sehr mächtige Funktion der Computeralgebra betrifft die Faktorisierung von Polynomen. Kann man ein Polynom `f` so in Faktoren zerlegen, daß keine algebraischen Ausdrücke wie Wurzeln entstehen, so findet das Kommando `Factor[f]` diese Faktorisierung. Wir faktorisieren den in Zeile 5 definierten ausmultiplizierten Term `f`.

```
In[6]:= Factor[f]
```

$$\text{Out[6]} = (-2x + y) (-x + y) (-2x + y) (-x + y) (-2x + y) (-x + y)$$

MATHEMATICA kann auch mit rationalen Funktionen rechnen. Die `Simplify`-Funktion versucht, Vereinfachungen vorzunehmen. Hier vereinfachen wir den Term

$$\frac{1}{1-x} - \frac{1+x}{1-x^2}$$

```
In[7]:= Simplify[1/(1-x)-(1+x)/(1-x^2)]
```

```
Out[7]= 0
```

Man beachte, daß rein algebraisch vereinfacht wird ohne Berücksichtigung von Nullstellen des Nennerpolynoms. Man kann rationale Funktionen faktorisieren, z.B. die Funktion

$$\frac{6 + 21x + 21x^2 + 6x^3}{-2x - 4x^2 + 6x^3}$$

durch den Befehl

```
In[8]:= g=Factor[(6+21x+21x^2+6x^3)/(-2x-4x^2+6x^3)]
```

$$\text{Out[8]} = \frac{3(1+x)(2+x)(1+2x)}{2(-1-3x)(1-x)x}$$

Mit `Apart` kann man die Partialbruchzerlegung einer rationalen Funktion berechnen lassen.

```
In[9]:= Apart[g,x]
```

$$\text{Out[9]} = 1 + \frac{27}{4(-1+x)} - \frac{3}{x} + \frac{5}{4(1+3x)}$$

Für das Lösen von Polynomgleichungen bis zum vierten Grad gibt es bekanntlich Lösungsformeln, die MATHEMATICA anwendet. Es können dabei sehr lange und komplizierte Ausdrücke entstehen, die so unübersichtlich sind, daß man mit ihnen i.a. nicht viel anfangen kann. Im Beispiel der Gleichung $x^3 + ax^2 - ax - 1 = 0$, die durch das Kommando

```
In[10]:= Solve[x^3+a x^2-a x -1==0,x]
```

```
Out[10]= {{x -> 1}, {x ->  $\frac{-1 - a + \text{Sqrt}[-1 + a] \text{Sqrt}[3 + a]}{2}$ },
> {x ->  $\frac{-1 - a - \text{Sqrt}[-1 + a] \text{Sqrt}[3 + a]}{2}$ }}
```

nach x aufgelöst wird, sind die Lösungen lesbar. Da der Operator = eine Zuweisung bewirkt, wird die Gleichheit mathematischer Ausdrücke mit Hilfe des Symbols == ausgedrückt.

MATHEMATICA kann auch lineare Gleichungssysteme lösen, z.B. das System der beiden Gleichungen

$$x + y = a \quad \text{sowie} \quad x - by = c,$$

das durch

```
In[11]:= Solve[{x+y==a,x-b y==c},{x,y}]
```

```
Out[11]= {{x ->  $-\frac{a b}{-1 - b} - \frac{c}{-1 - b}$ , y ->  $-\frac{a}{-1 - b} + \frac{c}{-1 - b}$ }}
```

nach den beiden Variablen x und y aufgelöst wird. Man beachte, daß auch bei diesem Beispiel keine numerische, sondern eine symbolische Lösung erzielt wurde.

Das Kommando Table erzeugt einen Vektor bzw. eine Matrix, z.B. definieren wir mittels

```
In[12]:= A=Table[1/(j+k),{j,8},{k,8}]
```

```
Out[12]= {{-, -, -, -, -, -, -, -}, {-, -, -, -, -, -, -, --},
           2 3 4 5 6 7 8 9    3 4 5 6 7 8 9 10

> {-, -, -, -, -, -, --, --}, {-, -, -, -, -, --, --, --},
   4 5 6 7 8 9 10 11    5 6 7 8 9 10 11 12

> {-, -, -, -, --, --, --, --}, {-, -, -, --, --, --, --, --},
   6 7 8 9 10 11 12 13    7 8 9 10 11 12 13 14

> {-, -, --, --, --, --, --, --}, {-, --, --, --, --, --, --, --}}
   8 9 10 11 12 13 14 15    9 10 11 12 13 14 15 16
```

die sogenannte Hilbert-Matrix, bei der am Kreuzungspunkt der j -ten Zeile und der k -ten Spalte der Wert $1/(j+k)$ steht. Möchte man diese Matrix in der üblichen quadratischen Form sehen, gibt man

```
In[13]:= MatrixForm[A]
```

```
Out[13]//MatrixForm= 1  1  1  1  1  1  1  1
-  -  -  -  -  -  -  -
2  3  4  5  6  7  8  9

1  1  1  1  1  1  1  1
-  -  -  -  -  -  -  --
3  4  5  6  7  8  9  10

1  1  1  1  1  1  1  1
-  -  -  -  -  -  --  --
4  5  6  7  8  9  10  11

1  1  1  1  1  1  1  1
-  -  -  -  -  --  --  --
5  6  7  8  9  10  11  12

1  1  1  1  1  1  1  1
-  -  -  -  --  --  --  --
6  7  8  9  10  11  12  13

1  1  1  1  1  1  1  1
-  -  -  --  --  --  --  --
7  8  9  10  11  12  13  14

1  1  1  1  1  1  1  1
-  -  --  --  --  --  --  --
8  9  10  11  12  13  14  15

1  1  1  1  1  1  1  1
-  --  --  --  --  --  --  --
9  10  11  12  13  14  15  16
```

ein. Beim numerischen Rechnen gibt es große Probleme beim Invertieren dieser Matrix. Da MATHEMATICA rational exakt arbeitet, tauchen diese Probleme mit MATHEMATICA nicht auf.

```
In[14]:= Inverse[A]
```

```
Out[14]= {{2592, -60480, 498960, -1995840, 4324320, -5189184, 3243240,
> -823680}, {-60480, 1587600, -13970880, 58212000, -129729600, 158918760,
> -100900800, 25945920}, {498960, -13970880, 128066400, -548856000,
> 1248647400, -1553872320, 998917920, -259459200},
> {-1995840, 58212000, -548856000, 2401245000, -5549544000, 6992425440,
> -4540536000, 1189188000},
> {4324320, -129729600, 1248647400, -5549544000, 12985932960,
> -16527551040, 10821610800, -2854051200},
```

```

> {-5189184, 158918760, -1553872320, 6992425440, -16527551040,
> 21210357168, -13984850880, 3710266560},
> {3243240, -100900800, 998917920, -4540536000, 10821610800, -13984850880,
> 9275666400, -2473511040},
> {-823680, 25945920, -259459200, 1189188000, -2854051200, 3710266560,
> -2473511040, 662547600}}

```

Warum macht die Numerik hier Probleme? Weil die Determinante von A

```
In[15] := Det[A]
```

```

Out[15]= -----
1
4702142622508202833251304734720000000

```

so klein ist, daß sie bei der Rechnung mit einer festen Gleitpunktarithmetik nicht von Null unterschieden werden kann! Korrekterweise ergibt $A^{-1} \cdot A$ bei MATHEMATICA aber

```
In[16] := Inverse[A] . A
```

```

Out[16]= {{1, 0, 0, 0, 0, 0, 0, 0}, {0, 1, 0, 0, 0, 0, 0, 0},
> {0, 0, 1, 0, 0, 0, 0, 0}, {0, 0, 0, 1, 0, 0, 0, 0},
> {0, 0, 0, 0, 1, 0, 0, 0}, {0, 0, 0, 0, 0, 1, 0, 0},
> {0, 0, 0, 0, 0, 0, 1, 0}, {0, 0, 0, 0, 0, 0, 0, 1}}

```

wieder die Einheitsmatrix.

Mit der Funktion `Limit` können Grenzwerte berechnet werden, z.B. $\lim_{x \rightarrow 0} \frac{\sqrt{1+x} - \sqrt{1-x}}{x}$.

```
In[17] := Limit[(Sqrt[1+x]-Sqrt[1-x])/x,x->0]
```

```
Out[17]= 1
```

Zum (partiellen) Differenzieren gibt es die Funktion `D`, z.B. berechnen wir

$$\frac{\partial}{\partial x} \left(\frac{1+x}{1-x} \right)^n$$

```
In[18] := D[((1+x)/(1-x))^n,x]
```

```

Out[18]= n (-----) (----- + -----)
1 + x -1 + n      1      1 + x
1 - x           1 - x      (1 - x)^2

```

In[19]:= Simplify[%]

$$\text{Out[19]} = \frac{2^n \left(\frac{1+x}{1-x} \right)^{-1+n}}{(-1+x)^2}$$

Erst das Simplify-Kommando bringt das Resultat auf eine möglichst einfache Form.

MATHEMATICAS Series-Kommando berechnet Taylorpolynome, z.B. die ersten 14 Potenzen des Taylorpolynoms der Funktion

$$\sqrt{\frac{e^x - 1}{x}},$$

In[20]:= Series[Sqrt[(E^x-1)/x],{x,0,14}]

$$\begin{aligned} \text{Out[20]} = & 1 + \frac{x}{4} + \frac{5x^2}{96} + \frac{x^3}{128} + \frac{79x^4}{92160} + \frac{3x^5}{40960} + \frac{71x^6}{12386304} + \frac{113x^7}{247726080} + \\ & > \frac{3053x^8}{118908518400} + \frac{x^9}{22649241600} + \frac{17x^{10}}{930128855040} + \frac{19x^{11}}{744103084032} + \\ & > \frac{935917x^{12}}{1218840851644416000} - \frac{20287103x^{13}}{43878270659198976000} - \frac{2452337x^{14}}{210615699164155084800} + \\ & > 0[x]^{15} \end{aligned}$$

ein Ergebnis, mit welchem der Autor (s. [3]) eine Vermutung von Robertson über die Nichtnegativität der auftretenden Koeffizienten widerlegt hat.

Beispiele für die Berechnung von Stammfunktionen sowie bestimmten Integralen sind

$$\int e^x \cos x \, dx \quad \text{bzw.} \quad \int_0^{\infty} e^{-x^2} \, dx .$$

In[21]:= Integrate[E^x*Cos[x],x]

$$\text{Out[21]} = \frac{e^x \cos[x]}{2} + \frac{e^x \sin[x]}{2}$$

In[22]:= Integrate[Exp[-x^2],{x,0,Infinity}]

$$\text{Out[22]} = \frac{\text{Sqrt}[\text{Pi}]}{2}$$

Das Integrieren ist wie die Faktorisierung eine besonders eindrucksvolle Fähigkeit der Computeralgebrasysteme, da hier die Berechnung mit Papier und Bleistift häufig sehr mühsam ist, und da es andererseits algorithmische Möglichkeiten zum Integrieren gibt, die der Handberechnung nicht zugänglich sind, von einem Programm wie MATHEMATICA aber genutzt werden können. So gelingen MATHEMATICA viele Integrationen, an denen wir selbst scheitern würden.

Die analytischen Fähigkeiten von MATHEMATICA gehen so weit, daß sogar eine Funktion DSolve zum Lösen gewöhnlicher Differentialgleichungen zur Verfügung steht. Z.B. kann man die Differentialgleichungen

$$nr(x)y'(x) - r'(x)y(x) = 0 \quad \text{bzw.} \quad y''(x) - 2y'(x) + y(x) = 0 \quad (y(0) = y_0, y'(0) = y_1)$$

durch

```
In[23] := DSolve[n r[x] y'[x]-r'[x] y[x]==0,y[x],x]
```

```
Out[23]= {{y[x] -> C[1] r[x] 1/n}}
```

```
In[24] := DSolve[{y''[x]-2y'[x]+y[x]==0,y[0]==y0,y'[0]==y1},y[x],x]
```

```
Out[24]= {{y[x] -> E^x (y0 - x y0 + x y1)}}}
```

nach der gesuchten Funktion $y(x)$ auflösen.

Schließlich möchten wir noch auf MATHEMATICAS eindrucksvolle graphische Fähigkeiten hinweisen, welche mit ausschlaggebend für die weite Verbreitung und den guten Ruf MATHEMATICAS sind.

Der folgende Aufruf ergibt eine Abbildung der ersten 5 Tschebyscheff-Polynome.

```
In[25] := Plot[Evaluate[Table[ChebyshevT[n,x],{n,1,5}]],{x,-1,1},AspectRatio->Automatic]
```

```
Out[25]= -Graphics-
```

Die entsprechende Abbildung entsteht in einem separaten Fenster, s. Abbildung 1.

Es gibt eine Fülle von Optionen, mit denen man graphische Darstellungen nach eigenem Geschmack konfigurieren kann. Will man mehr über den MATHEMATICA Befehl `Befehl` wissen, so hilft `??Befehl` weiter:

```
In[26] := ??Plot
```

```
Plot[f, {x, xmin, xmax}] generates a plot of f as a function of x from xmin to
xmax. Plot[{f1, f2, ...}, {x, xmin, xmax}] plots several functions fi.
```

```
Attributes[Plot] = {HoldAll, Protected}
```

```
Options[Plot] =
```

```
{AspectRatio -> GoldenRatio^(-1), Axes -> Automatic, AxesLabel -> None,
AxesOrigin -> Automatic, AxesStyle -> Automatic, Background -> Automatic,
ColorOutput -> Automatic, Compiled -> True, DefaultColor -> Automatic,
Epilog -> {}, Frame -> False, FrameLabel -> None, FrameStyle -> Automatic,
FrameTicks -> Automatic, GridLines -> None, MaxBend -> 10.,
PlotDivision -> 20., PlotLabel -> None, PlotPoints -> 25,
PlotRange -> Automatic, PlotRegion -> Automatic, PlotStyle -> Automatic,
Prolog -> {}, RotateLabel -> True, Ticks -> Automatic,
DefaultFont -> $DefaultFont, DisplayFunction -> $DisplayFunction}
```




Abbildung 1: Die ersten 5 Tschebyscheff-Polynome

Mit `Plot3D` kann man dreidimensionale

```
In[27] := Plot3D[Cos[(x^2+y^2)/4]/(3+x^2+y^2), {x, -5, 5}, {y, -5, 5}, PlotPoints->50]
```

```
Out[27]= -SurfaceGraphics-
```

und mit `ContourPlot` gar implizite Graphiken erzeugen, s. Abbildungen 2 und 3.

```
In[28] := ContourPlot[Sin[x]Sin[y], {x, -5, 5}, {y, -5, 5}, PlotPoints->50]
```

```
Out[28]= -ContourGraphics-
```

Auch hier stehen eine Fülle von Optionen zur Verfügung.




Abbildung 2: Die Oberfläche $z = \frac{\cos \frac{x^2+y^2}{4}}{3+x^2+y^2}$




Abbildung 3: Die Höhenlinien der Funktion $\sin x \sin y$

Alles in allem können wir also feststellen, daß man mit den eingebauten MATHEMATICA-Funktionen allerhand mathematische Umformungen durchführen kann. In der Praxis mit MATHEMATICA wird man allerdings feststellen, daß jede der vorhandenen Funktionen auch auf ihre Grenzen stößt, sei es aus Speichermangel oder auch Implementierungsmängeln, s. z.B. [2]. Andererseits sind die Fähigkeiten insgesamt so mächtig, daß man MATHEMATICA sicherlich mit Gewinn einsetzen können wird, wenn man entsprechende Vorsicht walten läßt und nicht uneingeschränkt jedem berechneten Ergebnis ohne Kontrolle Glauben schenkt.

In diesem Abschnitt haben wir nur einen Bruchteil der Mathematik-relevanten Möglichkeiten MATHEMATICAS vorgestellt. Diese stellen wiederum nur einen Bruchteil des gesamten vorrätigen Funktionsumfangs dar. Man kann mit Listen operieren (wie z.B. in LISP), blockstrukturierte Programme schreiben (wie z.B. in PASCAL oder C) sowie regelorientiert programmieren. Die zuletzt genannte Technik werden wir nun vorstellen.

2 Definition der Ableitungsfunktion

Wir wollen uns das Ziel setzen, die Differentiation in MATHEMATICA zu erklären. Dabei wollen wir natürlich **nicht** die eingebaute Differentiationsprozedur verwenden.

Zunächst sagt man sich: Da die Ableitung f' einer Funktion f bzgl. der Variablen x als Grenzwert

$$f'(x) = \lim_{a \rightarrow x} \frac{f(a) - f(x)}{a - x} \quad (1)$$

definiert ist, und da MATHEMATICA Grenzwerte berechnen kann, verwendet man einfach die Definition und ist fertig. Aber Achtung: Wie berechnet MATHEMATICA (intern) Grenzwerte? Oder anders gefragt: Wie berechnen **wir** denn Grenzwerte in der Praxis?

Der Ableitungsgrenzwert (1) ist immer vom Typus $0/0$, und solche Grenzwerte werden in der Praxis mit der de l'Hospital'schen Regel berechnet — also mit Hilfe von Ableitungen! Derartige Verfahren muß auch MATHEMATICA zur Grenzwertberechnung heranziehen, so daß die obige Definition der Ableitungsfunktion letztlich offenbar doch den eingebauten Differentiationsmechanismus heranziehen muß, was wir ja nicht wollten.

Also überlegen wir von neuem: Wie berechnen wir denn Ableitungen in der Praxis? Dies geschieht durch das Anwenden von **Regeln** und einiger spezieller Ableitungsergebnisse. Genau so — wie einem Schüler also — werden wir nun MATHEMATICA das Differenzieren (erneut) beibringen.

Als erstes behandelt man gewöhnlich die Differentiation der Potenzfunktionen.

```
In[1]:= diff[c_,x_]:=0 /; FreeQ[c,x]
```

```
In[2]:= diff[x_^n_,x_]:=n*x^(n-1) /; FreeQ[n,x]
```

In der ersten Zeile erklären wir, daß Konstanten die Ableitung 0 besitzen. Die Regel `diff[c,x]->0` wird nur angewandt, falls die Bedingung `FreeQ[c,x]` erfüllt ist, d.h. der Ausdruck `c` das Symbol `x` nicht enthält, `c` also konstant bzgl. `x` ist. Bedingungen für die Anwendung von Regeln werden hinter dem Zeichen `/;` angegeben. Daß die angegebene Regel — unter besagter Bedingung — für **beliebige** Symbole `c` und `x` Anwendung finden soll, wird durch die Verwendung der Symbolik `c_` und `x_` ausgedrückt. Soll eine Regel nur für ein **bestimmtes** Symbol `y` gelten, verwendet man dagegen kein Underscore-Zeichen `_`.

Schließlich bedeutet das `:=` Symbol wiederum eine Zuweisung, die allerdings im Gegensatz zur Zuweisung mittels `=` erst beim **Aufruf** der Funktion ausgeführt wird und bei der Definition von

Funktionen Verwendung findet. Daß keine augenblickliche Auswertung erfolgt, sieht man auch daran, daß keine Ausgabezeile `Out[n]` ausgegeben wird.

In der zweiten Zeile wird also definiert, daß die Ableitung von x^n den Wert nx^{n-1} erhalten soll, und zwar unter der Bedingung, daß n nicht von x abhängt. Mit dem Punkt beim Symbol `n_.` drücken wir aus, daß diese Regel auch noch für $n = 1$, wenn also explizit gar kein Exponent auftaucht, gültig sein soll.

Testen wir nun einmal, was MATHEMATICA gelernt hat:

```
In[3]:= diff[y,x]
```

```
Out[3]= 0
```

Richtig, y hängt ja nicht von x ab!

```
In[4]:= diff[x^(1/2),x]
```

```
Out[4]=  $\frac{1}{2 \sqrt{x}}$ 
```

Da wir vom Exponenten nicht verlangt hatten, daß er ganzzahlig ist, können nun also bereits beliebige Potenzen abgeleitet werden. Aber beim Beispiel

```
In[5]:= diff[2*x,x]
```

```
Out[5]= diff[2 x, x]
```

liefert MATHEMATICA als Ausgabe unsere Eingabe ab, da es nach Durchsicht aller Regeln, die wir für `diff` bisher aufgestellt haben, keine gefunden hat, die Anwendung finden könnte. Wir haben tatsächlich in der zweiten Zeile die Ableitungen lediglich für x^n erklärt, nicht für Vielfache davon. Dem kann aber leicht durch die Regel

```
In[6]:= diff[c_*f_,x_]:=c*diff[f,x] /; FreeQ[c,x]
```

abgeholfen werden. Nun kann MATHEMATICA das Problem von eben lösen:

```
In[7]:= diff[2*x,x]
```

```
Out[7]= 2
```

und hat überhaupt gelernt, daß man Konstanten vorziehen kann. Allerdings scheitert es an

```
In[8]:= diff[x+x^2,x]
```

```
Out[8]= diff[x + x2, x]
```

da wir noch nicht erklärt haben, wie mit einer Summe verfahren werden soll. Da die Differentiation linear ist, erklären wir also

```
In[9]:= diff[f_+g_,x_]:=diff[f,x]+diff[g,x]
```

mit dem Resultat, daß nun

```
In[10]:= diff[x+x^2,x]
```

```
Out[10]= 1 + 2 x
```

vereinfacht wird. Obwohl wir die Additivität nur für 2 Summanden definiert haben, wendet MATHEMATICA diese Regel nun auch auf mehrfache Summen an, da MATHEMATICA weiß, daß die Addition eine assoziative und kommutative Operation ist:

```
In[11]:= diff[Sum[k x^k,{k,0,10}],x]
```

```
Out[11]= 1 + 4 x + 9 x2 + 16 x3 + 25 x4 + 36 x5 + 49 x6 + 64 x7 + 81 x8 +  
> 100 x9
```

Ist andererseits bei einer Summe die obere (oder untere) Grenze keine ganze Zahl, kann die Regel nicht angewandt werden, da das Muster dann nicht mit dem einer endlichen Addition übereinstimmt.

```
In[12]:= diff[Sum[k x^k,{k,0,n}],x]
```

```
Out[12]= diff[Sum[k xk, {k, 0, n}], x]
```

Das beheben wir durch die Regel

```
In[13]:= diff[Sum[f_,{k_,k1_,k2_}],x_]:=Sum[diff[f,x],{k,k1,k2}]
```

```
In[14]:= diff[Sum[k x^k,{k,0,n}],x]
```

```
Out[14]= Sum[k x-1+k, {k, 0, n}]
```

Als nächstes wenden wir uns nun weiteren Ableitungsregeln zu. Bislang können z.B. keine Produkte abgeleitet werden

```
In[15]:= diff[(2x+x^2)(5+x^2-4x^3),x]
```

```
Out[15]= diff[(2 x + x2) (5 + x2 - 4 x3), x]
```

Daher teilen wir MATHEMATICA die Produktregel mit

```
In[16]:= diff[f_*g_,x_]:=diff[f,x]*g+diff[g,x]*f
```

und erhalten

```
In[17]:= diff[(2x+x^2)(5+x^2-4x^3),x]
```

```
Out[17]= (2 x - 12 x2) (2 x + x2) + (2 + 2 x) (5 + x2 - 4 x3)
```

Ebenso implementieren wir nun die Quotientenregel.

```
In[18]:= diff[f_/g_,x_] := (diff[f,x]*g-diff[g,x]*f)/g^2
```

Nun kann die Prozedur `diff` alle rationalen Funktionen ableiten.

```
In[19]:= diff[Sum[x^k,{k,1,10}]/Sum[k x^k,{k,1,5}],x]
```

```
Out[19]= ((x + 2 x2 + 3 x3 + 4 x4 + 5 x5)
> (1 + 2 x + 3 x2 + 4 x3 + 5 x4 + 6 x5 + 7 x6 + 8 x7 + 9 x8 + 10 x9) -
> (1 + 4 x + 9 x2 + 16 x3 + 25 x4)
> (x + x2 + x3 + x4 + x5 + x6 + x7 + x8 + x9 + x10)) /
> (x + 2 x2 + 3 x3 + 4 x4 + 5 x5)
```

Bevor wir MATHEMATICA nun noch die Kettenregel mitteilen, wollen wir die Ableitungen der elementaren Funktionen erklären:

```
In[20]:= diff[Log[x_],x_] := 1/x
```

```
In[21]:= diff[Sin[x_],x_] := Cos[x]
```

```
In[22]:= diff[Cos[x_],x_] := -Sin[x]
```

```
In[23]:= diff[Tan[x_],x_] := 1/Cos[x]^2
```

```
In[24]:= diff[Cot[x_],x_] := -1/Sin[x]^2
```

```
In[25]:= diff[Sec[x_],x_] := Sin[x]/Cos[x]^2
```

```
In[26]:= diff[Csc[x_],x_] := -Cos[x]/Sin[x]^2
```

```
In[27]:= diff[ArcSin[x_],x_] := 1/Sqrt[1-x^2]
```

```
In[28]:= diff[ArcCos[x_],x_] := -diff[ArcSin[x],x]
```

```
In[29]:= diff[ArcTan[x_],x_] := 1/(1+x^2)
```

```
In[30]:= diff[ArcCot[x_],x_] := -diff[ArcTan[x],x]
```

```
In[31]:= diff[ArcSec[x_],x_] := 1/(x^2*Sqrt[1-1/x^2])
```

```
In[32]:= diff[ArcCsc[x_],x_] := -diff[ArcSec[x],x]
```

```
In[33]:= diff[Erf[x_],x_] := (2/Sqrt[Pi])*E^(-x^2)
```

Hierbei ist Erf [x] die Fehlerfunktion, welche durch

$$\operatorname{erf}(x) := \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt$$

definiert ist und somit die angegebene Ableitung besitzt. Man sieht, daß es uns nicht schwerfallen würde, an dieser Stelle auch für weitere elementare Funktionen (wie z.B. für die hyperbolischen Funktionen oder MATHEMATICAS ExpIntegralEi, LogIntegral, SinIntegral sowie CosIntegral) Ableitungen festzusetzen.

Schließlich implementieren wir die Kettenregel. Bislang können ja z.B. die Ableitungen

```
In[34] := diff[Sin[ArcCos[x]], x]
```

```
Out[34]= diff[Sin[ArcCos[x]], x]
```

```
In[35] := diff[(1+x)^n, x]
```

```
Out[35]= diff[(1 + x)n, x]
```

nicht gefunden werden, obwohl das zweite Beispiel sogar eine rationale Funktion darstellt, falls $n \in \mathbb{N}$. Bei der Darstellung der Kettenregel müssen wir beachten, daß das Muster der Verkettung zweier Funktionen f_[g_] nicht alle verketteten Funktionen auffindet: Auch die Potenz f_^g ist eine verkettete Funktion, die aber nicht als solche erkennbar ist. Somit wird durch die beiden Regeln

```
In[36] := diff[f_[g_], x_] := diff[f[g], g] * diff[g, x]
```

```
In[37] := diff[f_g, x_] := fg * diff[g * Log[f], x]
```

die Kettenregel implementiert, und unsere obigen Beispiele liefern nun die Resultate

```
In[38] := diff[Sin[ArcCos[x]], x]
```

```
Out[39]= -(-----)
                2
            Sqrt[1 - x ]
```

```
In[40] := diff[(1+x)^n, x]
```

```
Out[40]= n (1 + x)-1 + n
```

Jetzt kann MATHEMATICA auch die Ableitung der Exponentialfunktion

```
In[41] := diff[Exp[x], x]
```

```
Out[41]= Ex
```



```

(* Potenzregeln *)
diff[c_,x_]:=0 /; FreeQ[c,x]
diff[x_^n_,x_]:=n*x^(n-1) /; FreeQ[n,x]

(* Linearitaet *)
diff[c_*f_,x_]:=c*diff[f,x] /; FreeQ[c,x]
diff[f_+g_,x_]:=diff[f,x]+diff[g,x]
diff[Sum[f_,{k_,k1_,k2_}],x_]:=Sum[diff[f,x],{k,k1,k2}]

(* Produktregel *)
diff[f_*g_,x_]:=diff[f,x]*g+diff[g,x]*f

(* Quotientenregel *)
diff[f_/g_,x_]:=diff[f,x]*g-diff[g,x]*f/g^2

(* Ableitungen der elementaren Funktionen *)
diff[Log[x_],x_]:=1/x
diff[Sin[x_],x_]:=Cos[x]
diff[Cos[x_],x_]:=-Sin[x]
diff[Tan[x_],x_]:=1/Cos[x]^2
diff[Cot[x_],x_]:=-1/Sin[x]^2
diff[Sec[x_],x_]:=Sin[x]/Cos[x]^2
diff[Csc[x_],x_]:=-Cos[x]/Sin[x]^2
diff[ArcSin[x_],x_]:=1/Sqrt[1-x^2]
diff[ArcCos[x_],x_]:=-diff[ArcSin[x],x]
diff[ArcTan[x_],x_]:=1/(1+x^2)
diff[ArcCot[x_],x_]:=-diff[ArcTan[x],x]
diff[ArcSec[x_],x_]:=1/(x^2*Sqrt[1-1/x^2])
diff[ArcCsc[x_],x_]:=-diff[ArcSec[x],x]
diff[Erf[x_],x_]:=2/Sqrt[Pi]*E^(-x^2)

(* Kettenregel *)
diff[f_[g_],x_]:=diff[f[g],g]*diff[g,x]
diff[f_^g_,x_]:=f^g*diff[g*Log[f],x]

(* Hauptsatz der Differential- und Integralrechnung *)
diff[Integrate[f_,x_],x_]:=f

```

die Differentiationsprozedur `diff` vollständig festlegt.

Wir halten diese Programmieretechnik, die sich an der üblichen mathematischen Vorgehensweise orientiert, für einen der wesentlichen Vorzüge von MATHEMATICA gegenüber anderen Systemen.

3 Schlußbemerkung

Wir wollen an dieser Stelle betonen, daß es berechtigte kritische Stimmen zur Konzeption von MATHEMATICA sowie ihrer Konkretisierung gibt, s. z.B. [2]. Andererseits bietet uns die Konzeption von MATHEMATICA die Möglichkeit, Probleme so zu formulieren, wie sie sich uns in unserer üblichen mathematischen Denkweise bereits stellen. Stellt man nicht allzu große Forderungen an die Tiefe der Probleme, mit denen man sich beschäftigen will, so ist MATHEMATICA als Hilfsmittel allererste Wahl.

Anschrift des Verfassers: Dr. Wolfram Koepf, Seelingstr. 21, 1000 Berlin 19

Literatur

- [1] Engel, A.: Eine Vorstellung von Derive. In: Didaktik der Mathematik **18** (1990), 165–182.
- [2] Fateman, R. J.: A review of Mathematica. J. Symbolic Computation **13** (1992), 545 – 579
- [3] Koepf, W.: On two conjectures of M. S. Robertson. Complex Variables **16** (1991), 127 - 130
- [4] Rich, A., Rich, J. und Stoutemyer, D.: *DERIVE User Manual*, Version 2, Soft Warehouse, Inc., 3660 Waiialae Avenue, Suite 304, Honolulu, Hawaii, 96816-3236.
- [5] Wolfram, St.: *MATHEMATICA. A system for doing mathematics by Computer*. Addison-Wesley Publ. Comp., Redwood City, Calif. 1991.